NAG Library Routine Document

G01FMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01FMF returns the deviate associated with the lower tail probability of the distribution of the Studentized range statistic, via the routine name.

2 Specification

```
FUNCTION G01FMF (P, V, IR, IFAIL)
REAL (KIND=nag_wp) G01FMF
INTEGER            IR, IFAIL
REAL (KIND=nag_wp) P, V
```

3 Description

The externally Studentized range, q, for a sample, $x_1, x_2, \ldots, x_r$, is defined as

$$q = \frac{\max(x_i) - \min(x_i)}{\hat{\sigma}_e},$$

where $\hat{\sigma}_e$ is an independent estimate of the standard error of the x_i. The most common use of this statistic is in the testing of means from a balanced design. In this case for a set of group means, $\bar{T}_1, \bar{T}_2, \ldots, \bar{T}_r$, the Studentized range statistic is defined to be the difference between the largest and smallest means, $\bar{T}_{\text{largest}}$ and $\bar{T}_{\text{smallest}}$, divided by the square root of the mean-square experimental error, MS_{error}, over the number of observations in each group, n, i.e.,

$$q = \frac{\bar{T}_{\text{largest}} - \bar{T}_{\text{smallest}}}{\sqrt{MS_{\text{error}}/n}}.$$

The Studentized range statistic can be used as part of a multiple comparisons procedure such as the Newman–Keuls procedure or Duncan's multiple range test (see Montgomery (1984) and Winer (1970)).

For a Studentized range statistic the probability integral, $P(q; v, r)$, for v degrees of freedom and r groups, can be written as:

$$P(q; v, r) = C \int_0^\infty x^{v-1} e^{-vx^2/2} \left(r \int_{-\infty}^{\infty} \phi(y)(\Phi(y) - \Phi(y - qx))^{r-1}\, dy \right) dx,$$

where

$$C = \frac{v^{v/2}}{\Gamma(v/2)2^{v/2-1}}, \quad \phi(y) = \frac{1}{\sqrt{2\pi}} e^{-y^2/2} \quad \text{and} \quad \Phi(y) = \int_{-\infty}^{y} \phi(t)\, dt.$$

For a given probability p_0, the deviate q_0 is found as the solution to the equation

$$P(q_0; v, r) = p_0, \tag{1}$$

using C05AZF . Initial estimates are found using the approximation given in Lund and Lund (1983) and a simple search procedure.

4 References

Lund R E and Lund J R (1983) Algorithm AS 190: probabilities and upper quartiles for the studentized range *Appl. Statist.* **32(2)** 204–210

Montgomery D C (1984) *Design and Analysis of Experiments* Wiley

Winer B J (1970) *Statistical Principles in Experimental Design* McGraw–Hill

5 Parameters

1: P – REAL (KIND=nag_wp) *Input*

On entry: the lower tail probability for the Studentized range statistic, p_0.

Constraint: $0.0 < \mathrm{P} < 1.0$.

2: V – REAL (KIND=nag_wp) *Input*

On entry: v, the number of degrees of freedom.

Constraint: $\mathrm{V} \geq 1.0$.

3: IR – INTEGER *Input*

On entry: r, the number of groups.

Constraint: $\mathrm{IR} \geq 2$.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if $\mathrm{IFAIL} \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\mathrm{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01FMF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If on exit $\mathrm{IFAIL} = 1$, then G01FMF returns 0.0.

$\mathrm{IFAIL} = 1$

On entry, $\mathrm{P} \leq 0.0$,
or $\mathrm{P} \geq 1.0$,
or $\mathrm{V} < 1.0$,
or $\mathrm{IR} < 2$.

$\mathrm{IFAIL} = 2$

The routine was unable to find an upper bound for the value of q_0. This will be caused by p_0 being too close to 1.0.

IFAIL $= 3$

There is some doubt as to whether full accuracy has been achieved. The returned value should be a reasonable estimate of the true value.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The returned solution, q_*, to equation (1) is determined so that at least one of the following criteria apply.

(a) $|P(q_*;v,r) - p_0| \leq 0.000005$

(b) $|q_0 - q_*| \leq 0.000005 \times \max(1.0, |q_*|)$.

8 Parallelism and Performance

Not applicable.

9 Further Comments

To obtain the factors for Duncan's multiple-range test, equation (1) has to be solved for p_1, where $p_1 = p_0^{r-1}$, so on input P should be set to p_0^{r-1}.

10 Example

Three values of p, ν and r are read in and the Studentized range deviates or quantiles are computed and printed.

10.1 Program Text

```
    Program g01fmfe

!     G01FMF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: g01fmf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: p, v, valq
      Integer                          :: ifail, ir
!     .. Executable Statements ..
      Write (nout,*) 'G01FMF Example Program Results'
```

```
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) '  P     V    IR    Quantile '
      Write (nout,*)

d_lp: Do
        Read (nin,*,Iostat=ifail) p, v, ir
        If (ifail/=0) Then
          Exit d_lp
        End If

!       Compute deviate
        ifail = -1
        valq = g01fmf(p,v,ir,ifail)
        If (ifail/=0) Then
          If (ifail/=3) Then
            Exit d_lp
          End If
        End If

!       Display results
        Write (nout,99999) p, v, ir, valq
      End Do d_lp

99999 Format (1X,F5.2,2X,F4.1,1X,I3,1X,F10.4)
    End Program g01fmfe
```

10.2 Program Data

```
G01FMF Example Program Data
0.95 10.0  5
0.3  60.0 12
0.9   5.0  4
```

10.3 Program Results

```
 G01FMF Example Program Results

   P     V    IR    Quantile

  0.95  10.0   5     4.6543
  0.30  60.0  12     2.8099
  0.90   5.0   4     4.2636
```